

电子计算机常用算法 BASIC程序

关 恒 慎 编

中国人民解放军海军第二炮兵学院

电子计算机常用法 BASIC 程序

关恒慎 編



中国人民
解放军 海軍第二炮兵学院

附件

电子计算机常用算法 BASIC 程序

关恒慎 编

*

中国人民解放军海军第二炮兵学院 / 出
海军第二炮兵学院工厂 印刷

*

开本787×1092 1/16 印张 $7\frac{5}{16}$ 字数320,000字

1981年9月印刷 印数1—2500册

内 容 题 要

本书用BASIO语言编写了103个算法程序。集有初等函数计算, 扞值与数值微分, 数值积分, 线代数计算, 代数方程及超越方程求解, 常微分方程数值积分, 拟合与平滑, 特殊函数, 富里叶级数逼近, 随机数产生, 线性规划等工程中常用的计算机方法程序。为帮助读者理解程序使用方法, 每个算法程序均附有例题及上机结果, 可作为解决工程问题的参考资料。

前 言

本书是在中国科学院沈阳计算技术研究所等单位编写的《电子计算机常用算法》一书的基础上编写的。现作如下说明：

1. BASIC语言无过程概念，本书的程序均以子程序的形式出现。使用时只要参照例题写出主程序，调用子程序即可。调用时所需给出的参数及特殊约定详见每个子程序的使用说明及例题。书中大部分子程序的标号约定由200开始，使用者可根据需要改动。

2.为节省文字，每个算法只编写了功能、使用说明、程序及例题四个部分，删去了方法概要部分，读者如需要可查原书该部分内容。每个程序的例题均在DJ S—130计算机上进行了试算，并给出结果。

3.由于BASIC语言不能定义三维以上数组，因此原书中算法14，15，16未编写BASIC程序，但为了与原书查对方便，序号未改动。另外，由于DJ S—130机舍入误差较大，因此算法45，78等例题结果误差较大仅供读者参考。

4.由于编者水平所限，对书中的算法推敲不够，书中会有不少错误，尤其在程序结构上不一定是最简的，望读者发现后，给以订正，以使其完善。

本书在编写过程中得到中国科学院有关同志的热情帮助，我院刘洪钧、曹丽等同志在程序调试过程中做了大量工作，在此表示衷心感谢：

编 者

一九八一年三月

目 录

第一章 初等函数和复数运算	(1)
1 初等函数的计算	(1)
2 复数的除法	(2)
3 e^z (z 为复数)	(2)
4 复数的模与平方根	(3)
5 复变量三角函数	(3)
6 复变量的 Γ 指函数	(5)
7 复变量自然对数	(5)
第二章 扞值与数值微商	(6)
8 线性扞值	(6)
9 一元三点扞值	(7)
10 埃特金扞值	(8)
11 爱尔米特扞值	(9)
12 三次样条函数扞值、微商和积分	(10)
13 二重抛物拟合扞值, 微商或积分	(13)
第三章 数值积分	(15)
17 定步长辛普生求积	(15)
18 变步长辛普生求积	(15)
19 自适应辛普生求积	(16)
20 切比雪夫求积	(18)
21 龙贝格求积	(19)
22 外推法数值求积	(20)
23 自动选步长的辛普生方法求二重积分	(22)
24 高斯法求多重积分	(24)
第四章 线代数计算	(26)
§1 线代数方程组的求解	(26)
25 高斯消去法	(26)
26 列主元高斯消去法	(27)
27 全主元高斯消去法	(30)
28 行主元约当逐行消去法	(31)
29 对称方程组的平方根法	(33)
30 对称正定方程组改进平方根法	(34)
31 带型方程组的列主元消去法	(36)
32 对称带型方程组的解法	(37)

33	大型稀疏对称正定矩阵方程组的解法	(39)
34	解三对角型方程组的追赶法	(40)
35	共轭斜量法	(41)
36	病态方程组的迭代解法	(44)
37	复线性方程组的列主元消去法	(46)
§2	行列式求值及矩阵求逆	(48)
38	利用三角化求实行列式的值	(48)
39	求对称正定矩阵行列式的值	(49)
40	行主元消去法求逆矩阵	(50)
41	全主元高斯——约当消去法求逆	(51)
42	对称正定矩阵的求逆	(53)
§3	特征值与特征向量的计算	(55)
43	求实对称矩阵的特征值及特征向量的雅可比方法	(55)
44	用豪塞豪德变换化对称阵为三对角阵	(58)
45	实对称三对角矩阵的QL方法	(63)
46	用二分法计算对称三对角阵的特征值	(67)
47	对称带型矩阵的三对角化	(69)
48	广义特征值 ($AX = \lambda BX$, $ABX = \lambda X$ 等) 问题的简化	(72)
49	化一般矩阵为赫申伯格型矩阵	(77)
50	求实赫申伯格型矩阵特征值的QR算法	(82)
51	用改进的LR算法求复赫申伯格矩阵的特征值	(85)
52	求复矩阵的特征值及特征向量	(89)
53	求实矩阵的特征值及特征向量	(95)
第五章	求解代数方程和超越方程	(102)
54	三次和四次方程的代数解法	(102)
55	对分区间套法	(105)
56	求高次代数方程全部实根	(106)
57	抛物线法求实函数的实零点	(108)
58	抛物线法求任意函数零点	(110)
59	贝尔斯特——牛顿联合迭代法解高次方程	(116)
60	下降法解非线性方程组	(119)
61	解非线性方程组的拟牛顿法 I	(120)
62	解非线性方程组的拟牛顿法 II	(122)
63	改进的牛顿法	(124)
第六章	常微分方程的数值积分	(127)
64	定步长维梯方法	(127)
65	定步长龙格——库塔方法	(128)
66	定步长基尔方法	(129)
67	定步长五阶单步方法	(130)

68	变步长单步方法 I	(132)
69	变步长单步方法 II	(134)
70	定步长预报——校正方法	(137)
71	定步长哈明方法	(139)
72	外推法	(142)
73	混合方法	(144)
74	病态方程组的数值积分	(148)
75	自动积分法	(151)
第七章 拟合与平滑		(155)
76	五点三次平滑	(155)
77	样条函数平滑	(156)
78	有理切比雪夫逼近	(159)
79	切比雪夫曲线拟合	(163)
80	最小二乘曲线拟合	(165)
81	最小二乘曲面拟合	(168)
第八章 特殊函数		(175)
82	Γ 函数	(175)
83	Γ 函数的自然对数	(176)
84	第一类和第二类完全椭圆积分	(177)
85	贝塞尔函数	(178)
86	正交多项式	(179)
87	正态分布函数	(181)
88	指数积分	(183)
89	正弦积分、余弦积分和弗莱斯那积分	(183)
第九章 其它		(186)
90	多项式及其导数计算	(186)
91	富里叶级数逼近	(187)
92	快速富氏变换	(189)
93	满足均匀分布的随机数的产生 I	(195)
94	满足均匀分布的随机数的产生 II	(196)
95	满足正态分布的随机数的产生	(197)
96	正态随机偏差	(197)
97	满足普阿松分布的随机数的产生	(198)
98	三角回归	(200)
99	方差的因素分析	(206)
100	线性规划问题解法	(206)
101	全整数线性规划问题	(209)
102	广义逆矩阵及线性方程组的最短最小二乘解	(211)
103	阻尼最小二乘法	(214)

104	广义逆法解非线性方程组.....	(217)
105	变尺度方法求函数极小.....	(220)

第一章 初等函数和复数运算

1 初等函数的计算

一 功能

本程序用连分式计算以下七个函数：

$\sin x$, $\cos x$, $\tan x$, e^x ,
 $\sinh x$, $\cosh x$, $\tanh x$ 中的任

一个。

二 使用说明

(1)调用本程序应给出：

X 自变量值。

N 控制精度变量。

P 函数类型控制变量 $\left\{ \begin{array}{l} 1 \\ 2 \\ 3 \\ 4 \\ 5 \\ 6 \\ 7 \end{array} \right.$ 时

$$y = \begin{cases} \sin x \\ \cos x \\ \tan x \\ e^x \\ \sinh x \\ \cosh x \\ \tanh x \end{cases}$$

当 P 为 1—7 以外的其它数时，则 y 被赋以 x 值。

(2)计算出的函数值存放于

y。

三 程序

```
200 IF P<=3 GOTO 230
210 LET R=X*X
220 GOTO 240
230 LET R=-X*X
240 LET F=4*N+2
250 FOR I=N TO 1 STEP -1
260 LET F=4*I-2+R/F
270 NEXT I
```

```
280 IF P<=3 GOTO 310
290 LET N1=P-3
300 GOTO 320
310 LET N1=P+1
320 IF N1=1 GOTO 380
330 IF N1=2 GOTO 400
340 IF N1=3 GOTO 420
350 IF N1=4 GOTO 440
360 LET Y=X
370 GOTO 450
380 LET Y=(F+X)/(F-X)
390 GOTO 450
400 LET Y=2*X*F/(F*F-R)
410 GOTO 450
420 LET Y=(F*F+R)/(F*F-R)
430 GOTO 450
440 LET Y=2*X*F/(F*F+R)
450 RETURN
```

四 例题

计算 0.25, 0.5, $\pi/4$ 的所有七个函数值，取 n=4，计算结果：

X	0.25	0.5	0.7853982
$\sin x$	.247404	.479426	.707106
$\cos x$	.968912	.877583	.707107
$\tan x$	.255342	.546303	.999999
e^x	1.28403	1.64872	2.19328
$\sinh x$	.252612	.521095	.86867
$\cosh x$	1.03141	1.12763	1.32461
$\tanh x$	.244919	.462117	.655794

主程序

```
5 LET N=4
```

```

10 LET X=.25
20 FOR P=1 TO 7
30   GOSUB 200
40   PRINT Y
50 NEXT P
51 PRINT
52 PRINT
53 PRINT
60 LET X=.5
70 FOR P=1 TO 7
80   GOSUB 200
90   PRINT Y
100 NEXT P
101 PRINT
102 PRINT
103 PRINT
110 LET X=3.14159/4
120 FOR P=1 TO 7
130   GOSUB 200
140   PRINT Y
150 NEXT P
160 END

```

2 复数的除法

一 功能

本程序用以求 $a+ib$ 和 $c+id$ 的商。

二 使用说明

(1)调用本程序时应给出:

A, B, C, D.

(2)结果实部在 E 中, 虚部在 F 中。

三 程序

```

200 IF ABS (C)>=ABS (D)
    GOTO 260
210 LET R=C/D
220 LET M=D+R*C
230 LET E=(A*R+B)/M
240 LET F=(B*R-A)/M
250 GOTO 300
260 LET R=D/C

```

```

270 LET M=C+R*D
280 LET E=(A+B*R)/M
290 LET F=(B-A*R)/M
300 RETURN

```

四 例题

计算:

$$Z = \frac{1.5_{10}^{10} + 10_{10}^{20}i}{2_{10}^{37} + 10_{10}^{29}i}$$

计算结果 $z = 2.575_{10}^{-26} + 0.5_{10}^{-18}i$

主程序 ($A = 1.5_{10}^{10}$, $B = 10_{10}^{20}$,

$C = 2_{10}^{37}$, $D = 10_{10}^{29}$)

```

10 INPUT A,B,C,D
15 PRINT
20 GOSUB 200
30 PRINT E,F
40 END

```

3 e^z (Z 为复数)

一 功能

本程序用以计算 e^z , 其中 $z = a+ib$ 。

二 使用说明

(1)调用本程序时应给出:

A Z 的实部。

B Z 的虚部。

(2)结果存放于

C 结果的实部。

D 结果的虚部。

三 程序

```

200 LET C=EXP (A)
210 LET D=C* SIN (B)
220 LET C=C* COS (B)
230 RETURN

```

四 例题

计算:

$$Z = e^{1+\frac{\pi}{4}i}$$

结果 $Z = 1.92212 + 1.92211i$

主程序 ($A = 1$, $B = 3.141593/4$)

```

10 LET A=1
20 LET B=3.14159/4
30 GOSUB 200
40 PRINT C, D
50 END

```

4 复数的模与平方根

一 功能

本部分提供两个程序，第一个程序用以求复数 $Z = x + iy$ 的模 $|Z|$ ，两个程序连用，用以求 Z 的平方根 $\sqrt{Z} = a + ib$ (取 $a > 0$)。

二 使用说明

(1)调用本程序时应给出：

X 给定复数的实部。

Y 给定复数的虚部。

(2)结果 $|Z|$ 存于 C 中，A、B 分别存放 $\sqrt{Z}$ 的实部和虚部。

三 程序

(1)求 $|Z|$

```

200 LET X=ABS (X)
210 LET Y=ABS (Y)
220 IF X=0 GOTO 270
230 IF Y=0 GOTO 290
240 IF X>Y GOTO 310
250 LET C=Y*SQR(1+((X/Y)
    *(X/Y)))
260 GOTO 320
270 LET C=X
280 GOTO 320
290 LET C=Y
300 GOTO 320
310 LET C=X*SQR(1+(Y/
    X)2)
320 RETURN

```

(2)求 $\sqrt{Z}$

```

400 IF X=0 GOTO 480
410 GOSUB 200
420 LET A=SQR ((ABS (X)

```

+ C)/2)

```

430 IF X>0 GOTO 530
440 IF Y<0 GOTO 470
450 LET B=A
460 LET A=Y/(B+B)
470 GOTO 570
480 IF Y=0 GOTO 500
490 GOTO 410
500 LET A=0
510 LET B=0
520 GOTO 570
530 LET B=Y/(A+A)
540 GOTO 570
550 LET B=-A
560 GOTO 460
570 RETURN

```

四 例题

计算 $Z = 0.1264_{10}^{15} + 0.1548_{10}^{15}i$ 的模和平方根。

计算结果：

$$|Z| = 1.9985_{10}^{14}$$

$$\sqrt{Z} = 1.2772_{10}^7 + 6.06011_{10}^6 i$$

主程序 (X = 0.1264₁₀¹⁵, Y = 0.1548₁₀¹⁵)

```

10 INPUT X, Y
15 PRINT
20 GOSUB 200
30 GOSUB 400
40 PRINT A, B, C
50 END

```

5 复变量三角函数

一 功能

本程序用以求 $\sin Z$, $\cos Z$, $\tan Z$ 及双曲函数 $\sinh y$, $\cosh y$, 其中 $Z = x + iy$ 。

二 使用说明

(1)调用本程序时应给出：

X 自变量实部。

Y 自变量虚部。

P 函数类型控制变量 = $\begin{cases} 1 \\ 2 \\ 3 \text{ 时,} \\ 4 \end{cases}$

相应于求 $\begin{cases} \sin(x+iy) \\ \cos(x+iy) \\ \tan(x+iy) \\ \sinh y, \cosh y \end{cases}$

(2) 计算结果存于

U sin, cos, tan 的实部或 coshy。

V sin, cos, tan 的虚部或 sinhy

三 程序

```
200 LET Y1=EXP (Y)
210 LET U=.5*(Y1+1/Y1)
220 IF ABS (Y)>= 1 GOTO
    410
225 DIM C(5)
230 LET C(0)=1.13032
240 LET C(1)=4.43368E-2
250 LET C(2)=5.42926E-4
260 LET C(3)=3.19844E-6
270 LET C(4)=1.10361E-8
280 LET C(5)=2.498E-11
290 LET B1=0
300 LET B2=0
310 LET Y1=2*(2*Y*Y-1)
320 FOR I=5 TO 0 STEP -1
330 LET B3=Y1*B1-B2-C
    (I)
340 IF I<>0 GOTO 360
350 GOTO 380
360 LET B2=B1
370 LET B1=B3
380 NEXT I
390 LET V=Y*(B3-B1)
400 GOTO 420
410 LET V=.5*(Y1-1/Y1)
```

```
420 IF P=1 GOTO 460
430 IF P=2 GOTO 490
440 IF P=3 GOTO 520
450 GOTO 660
460 LET U=U* SIN (X)
470 LET V=V* COS (X)
480 GOTO 660
490 LET U=U* COS (X)
500 LET V=-V* SIN (X)
510 GOTO 660
520 LET A=U* SIN (X)
530 LET B=V* COS (X)
540 LET C1=U* COS (X)
550 LET D=-V* SIN (X)
560 IF ABS (C1)>= ABS
    (D) GOTO 620
570 LET R=C1/D
580 LET N=D+R*C1
590 LET U=(A*R+B)/N
600 LET V=(B*R-A)/N
610 GOTO 660
620 LET R=D/C1
630 LET N=C1+R*D
640 LET U=(A+B*R)/N
650 LET V=(B-A*R)/N
660 RETURN
```

四 例题

求 $Z=0.25+0.25i$ 时的 $\sin Z$, $\cos Z$, $\tan Z$, $\sinh y$, $\cosh y$ 。

计算结果:

$\sin Z = 0.255175 - 0.24476i$
 $\cos Z = 0.999349 + 6.24973i \cdot 10^{-2}$
 $\tan Z = 0.23909 - 0.25987i$
 $\sinh y = -0.252613$
 $\cosh y = 1.03141$

主程序 ($X=0.25$, $Y=0.25$)

```
10 INPUT X, Y
15 PRINT
20 FOR P=1 TO 4
```

```

30 GOSUB 200
40 PRINT U, V
50 NEXT P
60 END

```

6 复变量的幂指数函数

一 功能

本程序用以求幂指数函数 $Z = u^t$ ，其中
 $u = a + ib$, $t = o + id$ 。

二 使用说明

(1) 调用本程序时应给出：

A, B 底的实部及虚部。
 C, D 指数的实部及虚部。
 N L_n 中 2π 的倍数，
 N = 0 时，求得乘方主
 值。

(2) 结果的实部与虚部分别保存于 X，
 Y 中。

(3) 当底数为 0 时，程序打印出 'F A
 I L' 后停机。

三 程序

```

200 IF A = 0 GOTO 290
210 LET P = 6.28318 * N + ATN
    (B/A)
220 IF A < 0 GOTO 240
230 GOTO 310
240 IF B >= 0 GOTO 270
250 LET P = P - 3.14159
260 GOTO 310
270 LET P = P + 3.14159
280 GOTO 310
290 IF B = 0 GOTO 370
300 LET P = 1.57079 * (SGN(B)
    + 4 * N)
310 LET R = .5 * LOG (A * A
    + B * B)
320 LET V = C * P + D * R
330 LET W = EXP (C * R - D * P)
340 LET X = W * COS (V)

```

```

350 LET Y = W * SIN (V)

```

```

360 GOTO 380

```

```

370 PRINT "FAIL"

```

```

380 RETURN

```

四 例题

计算 $Z = (1 + i)^{1 + i}$ 的乘方主值。

结果 $Z = 0.273958 + 0.5837i$

主程序 (A = 1, B = 1, C = 1, D = 1,
 N = 0)

```

10 INPUT A, B, C, D, N
15 PRINT
20 GOSUB 200
30 PRINT "Z = ", X, "+ ",
    Y, "I"
40 END

```

7 复变量自然对数

一 功能

本程序用以求 $L_n Z$ ，其中 $Z = a +$
 $ib \neq 0$ 。

二 使用说明

(1) 调用本程序时应给出：

A 自变量实部。
 B 自变量虚部。

(2) 结果的实部与虚部分别存于 C，
 D 中。

(3) 当自变量为 0 时，程序打印出 'F
 A I L' 后停机。

三 程序

```

200 IF A = 0 GOTO 460
210 LET E = .5 * A
220 LET F = .5 * B
230 IF ABS (E) < .5 GOTO
    280
240 LET C = ABS (.5 * E) + A
    BS (.5 * F)
250 LET D = .5 * E / C * E + .5 *

```

```

      F/C * F
260 LET C = .5 * (LOG(C) +
      LOG(D)) + LOG(SQR(8))
270 GOTO 330
280 IF ABS(F) < .5 GOTO
      300
290 GOTO 240
300 LET C = ABS(2 * A) + A
      BS(2 * B)
310 LET D = 8 * A / C * A + 8 * B /
      C * B
320 LET C = .5 * (LOG(C) +
      LOG(D)) - LOG(SQR
      (8))
330 IF A < > 0 GOTO 360
340 LET D = -ATN(A/B) +
      1.5708 * SGN(B)
350 GOTO 490
360 IF ABS(E) >= ABS
      (F) GOTO 380
370 GOTO 340
380 IF A > 0 GOTO 420
390 IF B >= 0 GOTO 440

```

```

400 LET D = ATN(B/A)
      - 3.14159
410 GOTO 490
420 LET D = ATN(B/A)
430 GOTO 490
440 LET D = ATN(B/A) +
      3.14159
450 GOTO 490
460 IF B = 0 GOTO 480
470 GOTO 210
480 PRINT "FAIL"
490 RETURN

```

四 例题

计算 $Z = -1$ 的自然对数。

计算结果 $\ln Z = -7.15256_{10}^{-7} + 3.14159i$

主程序 (A = -1, B = 0)

```

10 INPUT A, B
15 PRINT
20 GOSUB 200
30 PRINT "LOG(Z) = ",
      C, "+ ", D, "I"
40 END

```

第二章 扞值与数值微分

8 綫性扞值

一 功能

本程序从给定的 n 个扞值结点中选取最靠近扞值点 x 的相邻二个扞值结点, 用綫性扞值公式对 m 个一元列表函数进行成组扞值。

二 使用说明

(1) 调用本程序时应给出:

N 扞值结点个数。

M 扞值的函数个数。

T 扞值点。

X 数组 X(N), 存放给定的扞值结点。

Y 数组 Y(M, N), 存放结点处的 m 组函数值。

(2) 结果存于数组 F(M) 中。

(3) 本程序适用于单个函数扞值, 但如果将下面程序做如下改动更为适宜:

去掉 60~80 语句, 改为 LET F = y(I) + U * (y(I+1) - y(I)); 这时结果存于 F 中, y(N) 为结点函数值。

三 程序

```

200 FOR I = 1 TO N-2

```

```

210 IF T <= X(I+1) GOTO
    240
220 NEXT I
230 LET I = N - 1
240 LET U = (T - X(I)) / (X
    (I+1) - X(I))
250 FOR J = 1 TO M
260 LET F(J) = Y(J, I)
    + U * (Y(J, I+1) - Y(J,
    I))
270 NEXT J
280 RETURN

```

四 例题

根据函数表:

i	1	2	3	4
x_1	0.1	0.2	0.3	0.4
$f_1(x_1)$	0.0998	0.1987	0.2955	0.3894
$f_2(x_1)$	0.9950	0.9801	0.9553	0.9211

求 $f_1(x)$ 和 $f_2(x)$ 在 $x = 0.25$ 处的值。

计算结果:

$$f_1(0.25) = 0.2471$$

$$f_2(0.25) = 0.9677$$

主程序 ($N = 4, M = 2, T = 0.25$)

```

10 INPUT N, M, T
20 PRINT
30 DIM X(N), Y(M, N),
    F(N)
40 FOR I = 1 TO N
50 READ X(I)
60 NEXT I
70 FOR I = 1 TO M
75 FOR J = 1 TO N
80 READ Y(I, J)
85 NEXT J
90 NEXT I
100 GOSUB 200
110 FOR I = 1 TO M
120 PRINT F(I)

```

```

130 NEXT I
140 END
150 DATA .1, .2, .3, .4, .0998,
    .1987, .2955
160 DATA .3894, .995, .9801,
    .9553, .9211

```

9 一元三点插值

一 功能

本程序从给定的 N 个插值点中选取最靠近插值点 x 的相邻三个插值结点, 用拉格朗日三点插值公式对 m 个一元列表函数进行成组插值。

二 使用说明

(1) 调用本程序时应给出:

N 给定的插值点个数。 $N \geq 3$

M 插值的函数个数。

T 插值点。

X 数组 $X(N)$, 存放给定的插值结点, 要求 $X(i) < x(i+1)$, $i = 1, 2, \dots, N-1$ 。

Y 数组 $Y(M, N)$, 存放结点处的 m 组函数值。

(2) 结果存于数组 $F(M)$ 中。

(3) 本程序可用于单个函数插值, 若做如下改动更为适宜:

去掉120—140语句, 改为:

```

LET F = U * Y(I) + V * Y(I
+ 1) + W * Y(I + 2), 其中 F 为 简
变, Y 为一维数组 Y(N)。

```

三 程序

```

200 FOR I = 1 TO N - 3
210 IF T <= X(I+1) GOTO
    240
220 NEXT I
230 LET I = N - 2
240 IF I < > 1 GOTO 350
250 LET X0 = X(I)
260 LET X1 = X(I+1)

```

```

270 LET X2=X(I+2)
280 LET U=(T-X1)*(T-
    X2)/((X0-X1)*(X0-
    X2))
290 LET V=(T-X0)*(T-
    X2)/((X1-X0)*(X1-
    X2))
300 LET W=(T-X0)*(T-
    X1)/((X2-X0)*(X2-
    X1))
310 FOR J=1 TO M
320 LET F(J)=U*Y(J,
    I)+V*Y(J, I+1)+W
    *Y(J, I+2)
330 NEXT J
340 GOTO 390
350 IF T-X(I)<X(I+1)
    GOTO 370
360 GOTO 250
370 LET I=I-1
380 GOTO 250
390 RETURN

```

四 例题

根据函数表

i	1	2	3	4
x_1	0.1	0.2	0.3	0.4
$f_1(x_1)$	0.0998	0.1987	0.2955	0.3894
$f_2(x_1)$	0.9950	0.9801	0.9553	0.9211

求 $f_1(x)$ 和 $f_2(x)$ 在 $x=0.25$ 处的值。

计算结果： $f_1(0.25)=0.247362$ ，
 $f_2(0.25)=0.968937$ 。

主程序(N=4, M=2, T=0.25)

```

10 INPUT N, M, T
15 PRINT
20 DIM X(N), Y(M, N),
    F(M)
30 FOR I=1 TO N

```

```

40 READ X(I)
50 NEXT I
60 FOR I=1 TO M
70 FOR J=1 TO N
80 READ Y(I, J)
90 NEXT J
100 NEXT I
110 GOSUB 200
120 FOR I=1 TO M
130 PRINT F(I)
140 NEXT I
150 DATA .1, .2, .3, .4, .0998,
    .1987, .2955
160 DATA .3894, .995, .9801,
    .9553, .9211
170 END

```

10 埃特金扞值

一 功能

本程序从给定的N个扞值结点中取最靠近X的相邻m($m \leq n$)个结点，用埃特金反复线性扞值公式对一元函数进行扞值。

二 使用说明

(1)调用本程序时应给出：

- N 扞值结点个数。
- M 用以扞值的结点个数， $M \leq n$ 。
- T 扞值点。
- X 数组X(N)，存放扞值结点，要求 $X(i) < X(i+1)$ ， $i=1, 2, \dots, N-1$ 。
- Y 数组Y(N)，存放结点的函数值。

(2)结果存于A中。

三 程序

```

200 IF M>N GOTO 260
210 FOR I=1 TO N
220 IF T<=X(I) GOTO

```